*/ Files needed: LASAZ004 (sex), LASAC151 (bloodpressure), LASAC161 (height and weight and waist circumference), LASAC861 (Interleukin-6 and C-reactive protein), LASAC867 (total cholesterol/HDL cholesterol
and HDL cholesterol), LASAC860 (glucose)

*/ Can only be constructed for those who participated in the medical interview at C-wave (N=1509) and for whom blood was collected and biomarkers were
estimated (N=1352 out of N=1509).

*/ BEGIN WITH OPENING DATA FILE LASAZ004 AND MERGING NECESSARY FILES (SEE ABOVE)

FREQUENCIES VARIABLE = cmvar800 cmvar801 cmglucos cmhscrp cmil6 cmed150 cmed153 cmed156 cmed157 cmed159 cmed160 cmhdl2 cmchol2
 /ORDER=ANALYSIS.

*/ Glucose data at C-wave is available only for respondents from Zwolle and surroundings, not Oss or Amsterdam.

MISSING VALUES cmvar800 cmvar801 cmglucos cmhdl2 (-1,-2).
MISSING VALUES cmhscrp cmil6 (-1,-2,-3).
MISSING VALUES cmhdl2 (-1).

**********SYSTOLIC BLOEDDRUK**********
****

FREQUENCIES VARIABLES=cmvar800
 /NTILES=4
 /STATISTICS=STDDEV MINIMUM MAXIMUM SEMEAN MEAN MEDIAN
 /HISTOGRAM NORMAL
 /ORDER=ANALYSIS.

*/ 1473 with a valid score on systolic bp.
*/ percentile scores at: 134, 151, and 170.

**********DIASTOLIC**********
****

FREQUENCIES VARIABLES=cmvar801
 /NTILES=4
 /STATISTICS=STDDEV MINIMUM MAXIMUM SEMEAN MEAN MEDIAN
 /HISTOGRAM NORMAL
 /ORDER=ANALYSIS.

*/ 1473 with a valid score on distolic bp.
*/ percentilescores at: 74, 82, and 91.

**********GLUCOSE**********
****

```
FREQUENCIES VARIABLES=cmglucos
 /NTILES=4
  /STATISTICS=STDDEV MINIMUM MAXIMUM SEMEAN MEAN MEDIAN
  /HISTOGRAM NORMAL
  /ORDER=ANALYSIS.

*/ 404 with a valid score for glucose.
*/ This has to do with the fact that glucose was determined only at blood collected in Zwolle (not Oss, not Amsterdam).
*/ Glucose added to the construction of AL, but might also be removed because of many missing values.
*/ percentilescores op: 4,9; 5,6; and 6,8.

****************************************CRP*************************************
*********************************************************************************

FREQUENCIES VARIABLES=cmhscrp
  /NTILES=4
  /STATISTICS=STDDEV MINIMUM MAXIMUM MEAN MEDIAN
/HISTOGRAM NORMAL
  /ORDER=ANALYSIS.

*/ 1287 with a valid CRP score.
*/ percentilescores at: 1, 5; 3,2; and 6,5.

****************************************IL-6*************************************
*********************************************************************************



FREQUENCIES VARIABLES=cmil6
  /NTILES=4
  /STATISTICS=STDDEV MINIMUM MAXIMUM MEAN MEDIAN
  /HISTOGRAM NORMAL
  /ORDER=ANALYSIS.

*/ 1287 with a valid IL-6 score.
*/ percentilescores at: 1,1; 1,9; and 3,2.
*/ From all respondents, 89.7% had IL-6 levels below the detection limit of 5 pg/mL.Another strategy used in LASA is to dichotomize IL-6 levels below the detection limit of 5 pg/mL at the median.

*********************************cBMI*************************************************
*********************************************************************************

***cBMI***.
IF  (cmed153>10) and (cmed150>100) cBMI=cmed153 / ((cmed150 / 100) * (cmed150 / 100)).
EXECUTE.


FREQUENCIES VARIABLES=cBMI
/NTILES=4
  /STATISTICS=STDDEV VARIANCE MINIMUM MAXIMUM SEMEAN
```

```
  /HISTOGRAM NORMAL
  /ORDER=ANALYSIS.

*/ 1481 with a valid cBMI score.
*/ percentilescores at: 24,1; 26,6; 29,3.


**********************************WAIST-HIP RATIO*************************
*******************************************************************************

**** waist mean **.
IF  ((cmed156>10) and (cmed157>10)) waist_gem=(cmed156+cmed157)/ 2.
EXECUTE.

**** hip mean **.
IF  ((cmed159>10) and (cmed160>10)) hip_gem=(cmed159+cmed160)/ 2.
EXECUTE.

***waist-hip ratio**.
IF  ((waist_gem>10) and (hip_gem>10)) waist_hip_ratio=waist_gem / hip_gem.
EXECUTE.

FREQUENCIES VARIABLES=waist_hip_ratio
  /NTILES=4
  /STATISTICS=STDDEV VARIANCE MINIMUM MAXIMUM SEMEAN
  /HISTOGRAM NORMAL
  /ORDER=ANALYSIS.

*/ 1410 with a valid waist_hip_ratio.
*/ percentilescores at: 0,86; 0,93; 0,99.

****************************HDL**********************************
*********************************************************************


***split file hdl by sex**.
SORT CASES  BY sex.
SPLIT FILE LAYERED BY sex.

**FREQ hdl per sex**.
FREQUENCIES VARIABLES=cmhdl2
  /NTILES=4
  /STATISTICS=STDDEV MINIMUM MAXIMUM MEAN MEDIAN
  /HISTOGRAM NORMAL
  /ORDER=ANALYSIS.


*/633 with a valid score on cmhdl2 for men.
*/662 with a valid score on cmhdl2 for women.
*/ percentilescores men at: 0,96; 1,16; 1,44.
*/ percentilescores women at: 1,11; 1,37; 1,70
*/ With HDL cholesterol the lowest quartile is high risk.
*/ There is 0.2 level difference in HDL between men and women -> this is why quartiles are determined
separately for men and women.
```

```
SPLIT FILE OFF.


**********CHOLESTEROL/HDL RATIO**********
*******************************************************************************

****cholesterol/hdlratio**.
IF  ((cmchol2>0) and (cmhdl2>0)) chol_hdl_rat=cmchol2 / cmhdl2.
EXECUTE.

***split file chol/hdl ratio**.
SORT CASES  BY sex.
SPLIT FILE LAYERED BY sex.

FREQUENCIES VARIABLES=chol_hdl_rat
  /NTILES=4
  /STATISTICS=STDDEV VARIANCE MINIMUM MAXIMUM SEMEAN
  /HISTOGRAM NORMAL
  /ORDER=ANALYSIS.


*/633 with a valid score on chol/hdl ratio for men.
*/662 with a valid score on chol/hdl ratio for women.
*/ percentilescores men at: 3,59; 4,58; 5,76.
*/ percentilescores women at: 3,33; 4,14; 5,41.

SPLIT FILE OFF.


**********ALLOSTATIC LOAD**********
*******************************************************************

**********AL-systol**********

**AL-syst**.
IF  (cmvar800 > 1) AL_sys = 0.
EXECUTE.
IF  (cmvar800 > 170) AL_sys = 1.
EXECUTE.

*/ 75% percentilescores at: >170 mmHg.
*/ procedure for percentile use taken from T.L Gruenwald et al., L.D. Kubzansky et al.


**AL-diast**.
IF  (cmvar801>1) AL_dias=0.
EXECUTE.
IF  (cmvar801>91) AL_dias=1.
EXECUTE.

*/ 75% percentilescores at: >91 mmHg.
*/ procedure for percentile use taken from T.L Gruenwald et al., L.D. Kubzansky et al.
```

*/ use of syst en diast. bloodpressure as biomarker for AL taken from T. Seeman et al., T.L Gruenwald et al., L.D. Kubzansky et al.,G.Tanaka et al.

**********AL-Glucose**********

```
IF  (cmglucos>1) AL_gluc=0.
EXECUTE.
IF  (cmglucos>6.8) AL_gluc=1.
EXECUTE.
```

*/ 75% percentilescores at: >6.8 mMol/L.
*/ procedure for percentile use taken from T.L Gruenwald et al., L.D. Kubzansky et al.
*/ use of glucose as biomarker for AL taken from L.D. Kubzansky et al

**********AL_CRP**********

```
COMPUTE Al_CRP= -1.
IF  (cmhscrp>1) AL_CRP=0.
EXECUTE.
IF  (cmhscrp>6.5) AL_CRP=1.
EXECUTE.
```

*/ 75% percentilescores at: >6.5 ug/mL.
*/ procedure for percentile use taken from T.L Gruenwald et al., L.D. Kubzansky et al.
*/ use of CRP as biomarker for AL taken from T. Seeman et al.

**********AL_IL-6**********

```
COMPUTE Al_IL6= -1.
IF  (cmil6>1) AL_IL6=0.
EXECUTE.
IF  (cmil6>5) AL_IL6=1.
EXECUTE
```

*/ limit detecion value >5pg/mL.
*/ procedure for percentile use taken from T.L Gruenwald et al., L.D. Kubzansky et al.
*/ use of IL-6 as biomarker for AL taken from T. Seeman et al.

**********AL_cBMI**********

```
IF  (cBMI>1) AL_cBMI=0.
EXECUTE.
IF  (cBMI>29.3) AL_cBMI=1.
EXECUTE.
```

*/ 75% percentilescores at: >29.3 kg/m2.
*/ procedure for percentile use taken from T.L Gruenwald et al., L.D. Kubzansky et al.
*/ use of BMI as biomarker for AL taken from G.Tanakaet al

```
********************AL_WAIST-HIP RATIO ********

IF  (waist_hip_ratio>0.2) AL_waisHip=0.
EXECUTE.
IF  (waist_hip_ratio>0.99) AL_waisHip=1.
EXECUTE.

*/ 75% percentilescores at: >0.99.
*/ procedure for percentile use taken from T.L Gruenwald et al., L.D. Kubzansky et al.
*/ use of WAIST-HIP RATIO as biomarker taken from  G.Tanakaet al., T.L Gruenwald et al.

********************AL_HDL ********


IF (sex=1) and (cmhdl2>0.96) AL_hdl=0.
IF (sex=2) and (cmhdl2>1.11) AL_hdl=0.
IF (sex=1) and (cmhdl2<0.96) AL_hdl=1.
IF (sex=2) and (cmhdl2<1.11) AL_hdl=1.
EXECUTE.

*/ 75% percentilescores men at: <0,96 mmol/L.
*/ 75% percentilescores women at: <1,11 mmol/L.
*/ procedure for percentile use taken from T.L Gruenwald et al., L.D. Kubzansky et al.
*/ use of HDL as biomarker for AL taken from G.Tanakaet al., T.L Gruenwald et al., T. Seeman et al.

********************AL_ch_hdlratio ********

MISSING VALUES chol_hdl_rat (-1).

IF (sex=1) and (chol_hdl_rat<5.76) AL_ch_hdlrat=0.
IF (sex=2) and (chol_hdl_rat<5.41) AL_ch_hdlrat=0.
IF (sex=1) and (chol_hdl_rat>5.76) AL_ch_hdlrat=1.
IF (sex=2) and (chol_hdl_rat>5.41) AL_ch_hdlrat=1.
EXECUTE.

*/ 75% percentilescores men at: >5.76.
*/ 75% percentilescores women at: >5.41.
*/ procedure for percentile use taken from T.L Gruenwald et al., L.D. Kubzansky et al.
*/ use of chol_hdl_rat as biomarker for AL taken from G.Tanakaet al., T.L Gruenwald et al., T. Seeman et al.



*********************************************CATEGORIES AL**********************
*********************************************************************************
***

 *****cardio***************
*****************************

Compute valid_sys= 0.
IF AL_sys>=0 valid_sys =1.
Compute valid_dias= 0.
```

```
IF AL_dias>=0 valid_dias =1.
IF (valid_sys=1) and (valid_dias=1) cardiovasc= ( AL_sys + AL_dias)/2.
IF (valid_sys=1) and (valid_dias=0) cardiovasc= AL_sys.
IF (valid_dias=1) and (valid_sys=0) cardiovasc= AL_dias.
EXECUTE .

 *****inflammation*********************************
************************************************

Compute valid_IL6= 0.
IF AL_IL6>=0 valid_IL6 =1.
Compute valid_CRP= 0.
IF AL_CRP>=0 valid_CRP =1.
IF (valid_IL6=1) and (valid_CRP=1) inflam= ( AL_IL6 + AL_CRP)/2.
IF (valid_IL6=1) and (valid_CRP=0) inflam= AL_IL6.
IF (valid_CRP=1) and (valid_IL6=0) inflam= AL_CRP.
EXECUTE.

*******************metabolic**********************
*****************************************************

COMPUTE METAB = MEAN.3 (AL_cBMI, AL_waisHip, AL_hdl, AL_gluc, AL_ch_hdlrat).
EXECUTE.



**************ALLOSTATIC LOAD*****************
*****************************************************

COMPUTE AL = (metab + cardiovasc + inflam) .
EXECUTE.


FREQUENCIES VARIABLES AL.

*/ N=1169 valid scores for AL
```